


```

MOR_VERB:
    <mor perf> == "<stem perf>"
    <mor imperf> == "<stem imperf>" - .

% The basic stem is defined as the concatenation of the lexically %
% defined vowels and consonants. %

BASIC_STEM:
    <stem> ==
        "<stem cons1>" "<stem vowel1>" "<stem cons2>" "<stem vowel2>".

% Reduplication is defined at REDUP. The vowel of the initial %
% syllable <stem vowel1> is prefixed to the basic stem. The result %
% of this is then transduced at the node STEM_PHON. %

REDUP:
    <stem> == STEM_PHON:<"<stem vowel1>" BASIC_STEM>.

% The node STEM_PHON stands in for phonologically regular %
% alternations which accompany reduplication for the relevant %
% verbs in this fragment. The $maptoself variable is used to %
% map a character to itself. If the phonology were to be modelled %
% beyond the examples given here, we would need to include %
% variables for vowels and consonants, of course, rather than the %
% specific vowels and consonants of the lexical items included in %
% in this fragment. %

STEM_PHON:
    <> ==
        <$maptoself> == $maptoself <>
        <e e> == e ? e <>
        <a t> == a n d <>.

% Verb lexical entries. Verb1, Verb2 and Verb3 are regular: %
% ela-/e?ela 'jump', tik-/itik 'eat' and tana-/andana 'get %
% down'. Their telic/perfective stem is therefore the basic stem. %

Verb1:
    <> == VERB
    <gloss> == jump
    <stem vowel1> == e
    <stem cons2> == l
    <stem vowel2> == a.

Verb2:
    <> == VERB
    <gloss> == eat
    <stem cons1> == t
    <stem vowel1> == i
    <stem cons2> == k.

Verb3:
    <> == VERB
    <gloss> == get down
    <stem cons1> == t
    <stem vowel1> == a
    <stem cons2> == n
    <stem vowel2> == a.

% Verb lexical entries. Verb4, Verb5 and Verb6 are reversed %

```

```
% aspectual stems: taŋ/andaŋ- 'kick', nuŋ/ ʊnuŋ- 'pound', naŋ/anaŋ- %  
% 'cry'. Their telic/perfective stem is therefore the reduplicated %  
% stem. %
```

Verb4:

```
<> == POLAR_VERB  
<gloss> == kick  
<stem cons1> == t  
<stem vowel1> == a  
<stem cons2> == ŋ.
```

Verb5:

```
<> == POLAR_VERB  
<gloss> == pound  
<stem cons1> == n  
<stem vowel1> == ʊ  
<stem cons2> == ŋ.
```

Verb6:

```
<> == POLAR_VERB  
<gloss> == cry  
<stem cons1> == n  
<stem vowel1> == a  
<stem cons2> == ŋ.
```

```
# show  
<gloss>  
<syn cat>  
<mor perf>  
<mor imperf>.
```

```
# hide  
VERB  
POLAR_VERB  
MOR_VERB  
BASIC_STEM  
REDUP  
STEM_PHON.
```